Intégration numérique

2 Motivations

```
> f:=x->exp(-x^2);
  int(f(x),x=-1..1);
  I1:=evalf(int(f(x),x=-1..1));
```

$$f := x \to \mathbf{e}^{(-x^2)}$$

$$\mathrm{erf}(1)\sqrt{\pi}$$

$$I1 := 1.493648266$$

Pour calculer cette intégrale, Maple utilise la fonction d'erreur. Au vu de la définition de cette fonction, la réponse proposée n'est pas du tout explicite.

4 La méthode des rectangles

4.1 La méthode des rectangles à gauche

1.

```
> h=(b-a)/n;
```

$$h = \frac{b-a}{n}$$

2. Sg(f) devient :

```
> Sg(g)=h*sum(g(a[i]),i=0..n-1);
```

$$\mathrm{Sg}(g) = h\left(\sum_{i=0}^{n-1} \mathrm{g}(a_i)\right)$$

3.

```
> a[i]=a+i*h;
```

$$a_i = a + i\,h$$

4.

```
> rectangleg1:=proc(f,a,b,n)
                 local h;
  h:=evalf((b-a)/n);
  h*add(evalf(f(a+i*h)),i=0..n-1);
  end;
```

rectangleg1 := **proc**(f, a, b, n)

local h;

$h := \mathrm{evalf}((b-a)/n)$; $h * \mathrm{add}(\mathrm{evalf}(f(a+i*h)), i = 0 .. n-1)$

end proc

5.

```
> rectangleg1(f,-1,1,1000);
  I1;
```

$$1.493647775$$

$$1.493648266$$

6.

```
> rectangleg2:=proc(f,a,b,n)
                local M1,I1;
  M1:=max(seq(evalf(abs(D(f)(a+k*(b-a)/n))),k=0..n));
  I1:=rectangleg1(f,a,b,n);
  [I1-M1*(b-a)^2/(2*n),I1+M1*(b-a)^2/(2*n)];
  end;
```

rectangleg2 := **proc**(f, a, b, n)

local *M1*, *I1*;

M1 := max(seq(evalf(abs(D(f)($a + k*(b - a) / n$))), $k = 0 .. n$));

I1 := rectangleg1(f, a, b, n);

[*I1* − 1 / 2∗*M1*∗($b - a$)^2 / n, *I1* + 1 / 2∗*M1*∗($b - a$)^2 / n]

end proc

7.

```
> rectangleg2(f,-1,1,1000);
  I1;
```

[1.491932250, 1.495363300]

1.493648266

Remarque : On observe que I1 est bien dans l'intervalle proposé.

4.2 La méthode des rectangles à droite

1.

```
> rectangled1:=proc(f,a,b,n)
                local h;
  h:=evalf((b-a)/n);
  h*add(evalf(f(a+i*h)),i=1..n);
  end;
```

rectangled1 :=

proc(f, a, b, n) **local** h; h := evalf(($b - a$) / n); h∗add(evalf(f($a + i*h$)), $i = 1 .. n$) **end proc**

```
> rectangled1(f,-1,1,1000);
  I1;
```

1.493647775

1.493648266

Nous retrouvons le même résultat qu'avec la méthode des rectangles à gauche car la fonction f est paire.

2.

```
> rectangled2:=proc(f,a,b,n)
                local M1,I1;
  M1:=max(seq(evalf(abs(D(f)(a+k*(b-a)/n))),k=0..n));
  I1:=rectangleg1(f,a,b,n);
  [I1-M1*(b-a)^2/(2*n),I1+M1*(b-a)^2/(2*n)];
  end;
```

rectangled2 := **proc**(f, a, b, n)

local *M1*, *I1*;

M1 := max(seq(evalf(abs(D(*f*)(*a* + *k*∗(*b* − *a*) / *n*))), *k* = 0 .. *n*));

I1 := rectangleg1(*f*, *a*, *b*, *n*);

[*I1* − 1 / 2∗*M1*∗(*b* − *a*)^2 / *n*, *I1* + 1 / 2∗*M1*∗(*b* − *a*)^2 / *n*]

end proc

3.

```
> rectangled2(f,-1,1,1000);
  I1;
```

[1.491932250, 1.495363300]

1.493648266

Remarque : On observe que I1 est bien dans l'intervalle proposé.

5 La méthode des trapèzes

1.

```
> trapeze1:=proc(f,a,b,n);
  (rectangleg1(f,a,b,n)+rectangled1(f,a,b,n))/2;
  end;
```

trapeze1 :=

proc(*f*, *a*, *b*, *n*) 1 / 2∗rectangleg1(*f*, *a*, *b*, *n*) + 1 / 2∗rectangled1(*f*, *a*, *b*, *n*) **end proc**

2.

```
> trapeze1(f,-1,1,1000);
  I1;
```

1.493647775

1.493648266

Nous retrouvons la même approximation de I1 pour les mêmes raisons que précédemment.

3.

```
> trapeze2:=proc(f,a,b,n)
            local M2,I1;
  M2:=max(seq(evalf(abs(D[1$2](f)(a+i*(b-a)/n))),i=0..n));
  I1:=trapeze1(f,a,b,n);
  [I1-M2*(b-a)^3/12/n^2,I1+M2*(b-a)^3/12/n^2];
  end;
```

trapeze2 := **proc**(*f*, *a*, *b*, *n*)

local *M2*, *I1*;

M2 := max(seq(evalf(abs(D[1 $ 2](*f*)(*a* + *i*∗(*b* − *a*) / *n*))), *i* = 0 .. *n*));

I1 := trapeze1(*f*, *a*, *b*, *n*);

[*I1* − 1 / 12∗*M2*∗(*b* − *a*)^3 / *n*^2, *I1* + 1 / 12∗*M2*∗(*b* − *a*)^3 / *n*^2]

end proc

```
> trapeze2(f,-1,1,1000);
  I1;
```

[1.493646442, 1.493649108]

$$1.493648266$$

Remarque : Le résultat obtenu est le même dans les trois méthodes (car f est paire). Cependant, nous obtenonsun intervalle moins large par cette méthode.

4 La méthode de Simpson

1.

remarquons :

```
> (a[i]+a[i+1])/2=collect(simplify((a+i*h+a+(i+1)*h)/2),h);
```

$$\frac{1}{2}a_i + \frac{1}{2}a_{i+1} = \left(i + \frac{1}{2}\right)h + a$$

```
> simpson1:=proc(f,a,b,n)
            local h;
  h:=evalf((b-a)/n);
  h/6*( evalf(f(a)) + 4*add(evalf(f(a+(i+1/2)*h)),i=0..n-1) +
  2*add(evalf(f(a+i*h)),i=1..n-1) + evalf(f(b)) );
  end;
```

simpson1 := **proc**(*f*, *a*, *b*, *n*)

local *h*;

$h := \text{evalf}((b - a) / n);$

$1 / 6*h*(\text{evalf}(f(a)) + 4*\text{add}(\text{evalf}(f(a + (i + 1 / 2)*h)), i = 0 .. n - 1)$
$+ 2*\text{add}(\text{evalf}(f(a + i*h)), i = 1 .. n - 1) + \text{evalf}(f(b)))$

end proc

2.

```
> simpson1(f,-1,1,1000);
  I1;
```

$$1.493648265$$

$$1.493648266$$

On remarque que le résultat proposé est bien plus précis que celui des procédures précédentes.

3.

```
> simpson2:=proc(f,a,b,n)
            local M4,I1;
  M4:=max(seq(evalf(abs(D[1$4](f)(a+i*(b-a)/n))),i=0..n));
  I1:=simpson1(f,a,b,n);
  [I1-M4*(b-a)^5/(2880*n^4),I1+M4*(b-a)^5/(2880*n^4)];
  end;
```

simpson2 := **proc**(*f*, *a*, *b*, *n*)

local *M4*, *I1*;

$M4 := \max(\text{seq}(\text{evalf}(\text{abs}(D[1\ \$\ 4](f)(a + i*(b - a) / n))), i = 0 .. n));$

$I1 := \text{simpson1}(f, a, b, n);$

$[I1 - 1 / 2880*M4*(b - a)\text{^}5 / n\text{^}4, I1 + 1 / 2880*M4*(b - a)\text{^}5 / n\text{^}4]$

end proc

4.

```
> simpson2(f,-1,1,1000);
  I1;
```

$$[1.493648265, 1.493648265]$$

$$1.493648266$$

La précision des calculs n'est pas suffisante. Nous reprenons les calculs en augmentant la précision.

```
> Digits:=15;
  simpson2(f,-1,1,1000);
  evalf(int(f(x),x=-1..1));
```

$$\mathit{Digits} := 15$$

$$[1.49364826562474, 1.49364826562500]$$

$$1.49364826562486$$

```
>
```